
RPG: The Power of  
the Prototype

Everyone who does free format RPG logic needs to know at least the basics of using prototypes to replace their old 
parm lists for program calls. However many RPGers don't understand just how much prototypes can do for them.  
In this session, Susan reviews prototype basics but concentrates mostly on exploiting the often overlooked features 
of prototypes. Did you know that:  
 - Prototypes can protect you against parameter mis-matches 
 - Prototypes can provide automatic conversion of parameter types and sizes 
 - Prototypes can enable passing literals, built-in functions or function calls as parameters 
 - Prototypes allow you to specify optional parameters  
 - Prototypes enable you to call C functions and system APIs that return values 
We'll also look at how to exploit these features of prototypes effectively, including examples of the right way for called 
programs (or procedures) to deal with optional parameters. We'll cover the difference between passing parameters 
by value vs by reference and how to deal with very large return values from procedures. We'll talk about who should 
write prototypes and why they are often coded in separate members to be copied into RPG programs.  

The author, Susan Gantner, is co-founder of Partner400, a firm specializing in customized education and mentoring 
services for IBM i (AS/400 and iSeries) developers. After a 15 year career with IBM, including several years at the 
Rochester and Toronto laboratories, Susan is now devoted  to educating developers on techniques and technologies 
to extend and modernize their applications and development environments.  This is done via on-site custom classes 
for individual companies as well as conferences and user group events. Susan is also a partner in System i 
Developer, a consortium of IBM i educators who host events such as the RPG & DB2 Summit.  
Susan and her partner, Jon Paris, author regular technical articles for the IBM publication, IBM Systems Magazine, 
IBM i edition, and its companion electronic newsletter, IBM i EXTRA. You may view articles in current and past 
issues and/or subscribe to the free newsletter or the magazine on the web. You can also view Jon and Susan's blog 
at ibmsystemsmag.com/Blogs/iDevelop/. 
This presentation may contain small code examples that are furnished as simple examples to provide an illustration.  
These examples have not been thoroughly tested under all conditions.  We  therefore, cannot guarantee or imply 
reliability, serviceability, or function of these programs. 
All code examples contained herein are provided to you "as is".  THE IMPLIED WARRANTIES OF 
MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE EXPRESSLY DISCLAIMED.

Agenda
Prototyping basics 
• Basic syntax and usage 
• How prototypes catch parameter mismatches 
• How prototypes can fix mismatches for you (Constant parameters) 

Making calls more flexible 
• Pass constants, literals, expressions as parameters 
• Allow for optional parameters 

✦ 2 different kinds of optional parameters 
✦ How to code the called routine to properly handle the optional parameters 

Best practices for prototypes  
• Who should write them?  

✦ Where should they store them?  

• Why to use them in situations where the compiler may not technically 
require them (as of V7.1) 

As time permits ...  
• Using prototypes to call C or C-style APIs

We're going to focus primarily on using prototypes for our RPG code and the ways to make prototypes make your life 
easier.  

Near the end, we'll talk about how to prototype C functions and C-style APIs so that you can use them from RPG. 
There are a few examples in the handout. We will likely not have time to go through all of them in detail so there are 
also many notes alongside the charts to help you out if you find you want to try some of the examples on your own. 

Why Prototype Calls ?
Many reasons - some are better than others! 
• It's required to do calls in free format logic 
• Prototypes help detect parameter mismatches at compile time 
• Automatic conversion of parameter data types/sizes if necessary 
• Enables optional parameters 
• Enables calling of bound procedures and functions that return values 

Although Prototypes were introduced as part of the support for subprocedures, their utility goes way beyond that. Prototypes can 
protect you against parameter mis-matches and even provide for automatic conversion of parameter types and sizes. 
Prototyped calls are also the only way to do program calls in free format and therefore an essential tool for all modern RPG 
programmers. 
One of the really annoying problems with RPG III is that errors with parameter lists are not discovered until run-time.  It would be 
much better if we could have the compiler validate the parameter lists for us.   

Prototypes were added to the RPG IV language in the V3R2 and V3R6 releases.  Although their initial purpose was to support 
Prototyping of Subprocedures (more on these later) the RPG  developers realized that they could be used to provide this 
support.  As you'll see they also provided a number of other capabilities as well. 

The type of error that can result from the incorrect use of parameters can often cause RPG programmers sleepless nights and a 
difficult debugging tasks. 
For example: 
A 20 character field was expected but only a 10 character field was passed. 
The called program is modified and now moves blanks to the field instead of the 10 character field it used to move in.  This 
results in a corruption of data in the calling program.   
A character field is expected but a signed numeric is passed. 

A situation like this can continue for years since the internal representation of a character "1" is identical to the representation of 
the value 1 in a signed numeric field. A programmer makes a "simple change" and passes a different numeric parameter.  
Suddenly the called program starts producing incorrect output.  Why? 
Because the new parameter while numeric is a packed field and the representation of the digit 1 is now completely different 
causing the called program to fail to recognize it correctly. 

In this section we will look at some of the powerful options that prototypes give us to deal with situations such as these.  The 
main thing to note is that because the compiler is now able to validate parameters on the call statement, errors of the kind 
described above are very unlikely to occur 

Note: Some people mistakenly believe that prototypes and CALLP relate only to procedures.  This is not true.  CALLP can 
invoke either a program or a procedure.  As we shall see shortly, which type of call is made depends on keywords on the PR line 
of the prototype.

Anatomy of a Prototyped Call 

  

   D TaxCalc         PR                     ExtPgm('PX027C') 
   D  Gross                         8S 2 
   D  Allow                         8S 2 
   D  Tax                           8S 2 

  

     CallP TaxCalc( Gross : Allow : Tax); 

  

     Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
        GrossPay    Zoned(8:2); 
        Allowances  Zoned(8:2); 
        Tax         Zoned(8:2); 
     End-PR;

Name of the 
Prototype 

The name used on 
the call

Declares entry 
as a PRototype 

Name of the actual 
pgm/procedure to be called 

It's CaSe SeNsItIvE

ParametersPrototype Name 
(CALLP optional in free format)

V7.1+

Pre-V7.1

There are two main parts to a prototype.  
The first line is the declaration of the PR itself - either via Dcl-PR or a PR in the column where DS might otherwise 
go. This supplies the name of the program or procedure to be called. It also marks the beginning of the parameter 
list. If this were a bound procedure that returns a value, the return value could also be defined here.  
The second and subsequent lines describe the parameters to be passed to the called routine in sequence. 
The end of the parameter list is identified by the End-PR in free form or by the appearance of some other type of D 
spec entry (for example, DS, S or another PR) in the same column as the PR.  
Parameters are identified by a blank in positions 24-25  just as the subfields of a data structure are. 

Note that the names used for the parameters in the prototype do not need to match the names of the fields used on 
the call.  
In fact they could be completely blank and the compiler would be quite happy 
It only cares about the number and type of parameters - the names used are irrelevant 
Some people use a standard whereby the name used in the prototype identifies the type of field 

✦ e.g. Currency, Integer, Name, etc. 

The CALLP op code itself is a free-form operation.  The parameters are enclosed in parentheses and separated by 
colons. 
If there are no parameters, then the  parentheses can be omitted except in /Free-form logic where an empty set of 
parentheses must be coded.  The empty parentheses method preferred as it makes the intent of the code more 
obvious. 

     Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
        GrossPay    Zoned(8:2); 
        Allowances  Zoned(8:2); 
        Tax         Zoned(8:2); 
     End-PR; 
  

    TaxCalc( Gross : Allow : Tax); 

Basic Prototype Example 
If we are using these field definitions: 

Then the following CALL sequence:  

Can be replaced by this Prototype and CALLP:

 C                Call      'PX027C' 
 C                Parm                  Gross  
 C                Parm                  Allow  
 C                Parm                  Tax 

    Dcl-S    Gross    Zoned(8:2); 
    Dcl-S    Allow    Zoned(8:2); 
    Dcl-S    Tax      Zoned(8:2); 

With a prototype in place, any difference between the call statement and the prototype regarding the: 
✦ Number of parameters or the data type and size of a parameter 

will result in an error being signalled by the compiler.  The actual error is "The type and attributes of parameter 1 do 
not match those of the prototype". 
Note that it's not the field name difference (Gross vs GrossPay) that may be an issue. It's the data type and length. 
Numeric Parameters: The characteristics of the field must match exactly (type, size and decimal places).  
Character Parameters: 

✦ For varying length fields, the maximum length (i.e. the specified field length) must match exactly. 
✦ Fixed length character fields, the parameter passed must be at least as large as the length specified.  So if a 

larger field is passed that is OK, but a smaller field will be rejected. 
Any variance from these rules will be noted and immediately rejected by the compiler.  Compare this to a 
conventional CALL/PARM situation where the compiler would not produce an error, and the brand new bug could 
well be introduced into production. 
There are compiler options such as the keywords CONST and OPTIONS(*VARSIZE) which can be used to instruct 
the compiler to relax these rules somewhat.  We will deal with them later.    
Fixed format definitions for those not coding to V7.1+ standards:  

 D Gross            S               8S 2 
 D Allow            S               8S 2 
 D Tax              S               8S 2 

D TaxCalc         PR                     ExtPgm('PX027C') 
D  GrossPay                        8S 2 
D  Allowances                      8S 2 
D  Tax                             8S 2 
      CallP TaxCalc( Gross : Allow : Tax);

    Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
      GrossPay    Zoned(8:2); 
      Allowances  Zoned(8:2); 
      Tax         Zoned(8:2); 
    End-PR; 
  

    Dcl-S    Gross    Packed(8:2); 
    Dcl-S    Allow    Zoned(8:2); 
    Dcl-S    Tax      Zoned(8:2); 

      TaxCalc( Gross : Allow : Tax); 

The type and attributes of parameter 1 do not match those of the prototype 

Parameter Mis-match 
The compiler catches any parameter mismatches 
• Numeric fields must match prototype definition exactly 

✦ Size, decimal positions and type 
• Character fields much be same length or longer than prototype 
• However, the name of the field passed as a parameter is irrelevant 

Better to find this problem at compile time than run time!  
• Adding a keyword can help accommodate these mismatches

In our example the prototype has been hard coded in the calling program.   
Normally we would expect to see it being /COPY'd in from a source file supplied by the programmer who wrote PX027C.  
This is good practice since who knows better what parameters the program is expecting? 
This approach is going to seem like more work at first glance.   
It is!   
But there are two things it is important to remember: 
First - you only have to code it once. 
Second - our aim is not to speed the writing of the code, rather it is to reduce the opportunity for errors to be introduced 
during maintenance.  Since 80% of the effort expended on most programs is in maintenance/enhancements, the extra time 
pales into insignificance. 
Any variance from these rules will be noted and immediately rejected by the compiler.  Compare this to a conventional 
CALL/PARM situation where the compiler would not produce an error, and the brand new bug could well be introduced into 
production. 
On the next chart we will look at how the compiler reports such failures and the notes page describes the specific 
conditions the compiler checks for. 
For those still stuck with fixed format declarations:  

D TaxCalc         PR                     ExtPgm('PX027C') 
D  GrossPay                        8S 2 
D  Allowances                      8S 2 
D  Tax                             8S 2 

D Gross           S                9P 2 
D Allow           S                8S 2 
D Tax             S                8S 2 

    Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
      GrossPay    Zoned(8:2) Const; 
      Allowances  Zoned(8:2) Const; 
      Tax         Zoned(8:2); 
    End-PR; 
  

    Dcl-S    Gross     Packed(7:2); 
    Dcl-S    Base      Packed(6:2); 
    Dcl-S    PersAllow Packed(5:2); 
    Dcl-S    Tax       Zoned(8:2); 
  
     TaxCalc( Gross : Base + PersAllow : Tax); 

    //  Temp1 = Gross; 
    //  Temp2 = Base + PersAllow; 
    //  TaxCalc( Temp1 : Temp2 : Tax );    

Accommodating Parm Mismatches
Let the compiler help you - specify the keyword CONST 
• Implies that the called program will not modify the field 

✦ i.e. It will treat the parameter as Read-only 
✦ Do not specify CONST on any parameters that will be updated by the caller 

Allows the compiler to generate temporary fields when needed 
• For example if the size and/or type of the parameter don't match exactly  

✦ It also allows an expression or literal to be used instead of a variable

Compiler 
generates code 
similar to this 

The use of the CONST keyword allows the compiler to accommodate mismatches in the definition of the parameters 
between the callee and the caller.  
For example, the program we are calling expects a packed decimal value of five digits with no decimal places, but 
the field we would like to use as a parameter is a three digit signed numeric. 
Normally we would have to create a temporary variable (packed - five digits) and move the three digit number to it.  
We would then pass the temporary field as the parameter.  In fact the PARM op-code provides this support though 
the use of Factor 2. 
When you use the CONST keyword, you are telling the compiler that it is OK for it to make a copy of the data prior to 
passing it to accommodate such mismatches.  This avoids the need for us to explicitly define a working variable of 
the correct size and type. 
The compiler can accommodate differences in: 
 Size, decimal places, and type (zoned, packed, integer, etc.) - for Numeric fields  
 Date format - for Date fields only, obviously 
 Length and type (fixed or varying) - for Character fields 

Another benefit of using CONST is that it also allows an expression or a literal to be passed as a parameter. 
e.g. Count + 1     or  ‘Name’    or   32 

Fixed format version:

 D TaxCalc         PR                     ExtPgm('PX027C') 
 D  GrossPay                        8S 2  CONST 
 D  Allowances                      8S 2  CONST 
 D  Tax                             8S 2 
 :     :                             :  
 D Gross           S                7P 2 
 D Base            S                6P 2 
 D PersAllow       S                5P 4 
 D Tax             S                8S 2 
 :     :                             : 
     TaxCalc( Gross : Base + PersAllow : Tax); 

    Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
      GrossPay    Zoned(8:2) Const; 
      Allowances  Zoned(8:2) Const; 
      Tax         Zoned(8:2); 
    End-PR; 
  

    Dcl-S    Gross     Packed(7:2); 
    Dcl-S    GrossPay  Zoned(8:2); 
    Dcl-S    Allow     Zoned(8:2); 
    Dcl-S    Tax       Zoned(8:2); 

     TaxCalc( Gross : Allow : Tax ); // Temp generated for Gross (but not Allow) 

     TaxCalc( GrossPay : Allow : Tax);  // No temp generated here

Constant Parameters
The CONST keyword is a statement of Read-only intent 
• There is no guarantee that the called program will honor the intent 
• Temporary fields are only generated when needed 

✦ The original field is passed if no temporary field is required 
To guarantee Read-only define a matching Procedure Interface (PI) 
• Only then can the compiler ensure that the request is honored 
• The PI replaces the *ENTRY PLIST 

✦ We’ll see this in moment

One thing you have to be aware of is that the use of the keyword CONST in the prototype in the calling program is only a statement 
of intent by the calling program, promise - not a guarantee.  The called program could still change the value of the field.  You can 
only guarantee the read-only status of the parameter by also coding the CONST keyword on a Procedure Interface (PI), we will look 
at how to do this next.   
The use of the CONST keyword differs from using Factor 2 on a PARM op-code in that: 
  When Factor 2 is used, the value in that field is always copied to the result field before the call is made 
  When CONST is used, the parameter is only copied if there is a mismatch in the parameter 
Using the Factor 2 approach requires either that the programmer code a separate CALL/PARM combination for each field to be used 
as the parameter  or  hand code a move for each such field into the parameter field. 
While this may not seem like a big deal, any extraneous logic (such as Moves just to set up a parameter) obscure the real intent of 
the code and make it far harder for a maintenance programmer to understand what is going on. 

For fixed form declaration coders:  

 D TaxCalc         PR                     ExtPgm('PX027C') 
 D                                  8S 2  CONST 
 D                                  8S 2  CONST 
 D                                  8S 2 
 :       :                           :  
 D Gross           S                7P 2 
 D Allow           S                8S 2 
 D GrossPay        S                8S 2 
 D Tax             S                8S 2 

     TaxCalc( Gross : Allow : Tax );    // Temp field generated here 

     TaxCalc( GrossPay : Allow : Tax);  // No temp field generated here

    Dcl-PI    *N; 
      GrossPay    Zoned(8:2) Const; 
      Allowances  Zoned(8:2) Const; 
      Tax         Zoned(8:2); 
    End-PI; 
  

    Dcl-S    TotalPay  Packed(10:2); 

     TotalPay = GrossPay + Allowances; 

     GrossPay = 0;     // Error - Cannot change constant parameter

Using a Program PI with CONST
CONST means constant so compiler will issue an error if: 
• Any data is moved to a parameter described as CONST 

✦ Or any attempt is made to use the address of such a parameter 
- Such as passing it to another program that does not specify CONST 

• Names are typically omitted on PI 
✦ Since it refers to the program/procedure where it is coded, the name is assumed

    Dcl-PR    TaxCalc     ExtPgm('PX027C'); 
      GrossPay    Zoned(8:2) Const; 
      Allowances  Zoned(8:2) Const; 
      Tax         Zoned(8:2); 
    End-PR; 

Prototype in the caller 
(Best practice is to /Copy this, not 

hard coded)

Parameter names 
in PI are "real" unlike in the 

PR where they are 
documentary only 

If we replace the *ENTRY PLIST by placing a Procedure Interface and Prototype in the called program, the read-only intent will be 
guaranteed by the compiler. 
The compiler will issue an error if: 
  We attempt to modify the content of the field directly  
  We attempt to obtain its address (i.e. using %Addr). The reason for this is that once we have the address of a variable we can base 
another field on it and change the contents indirectly 
  We attempt to pass the field as a parameter to another program or procedure where it is not defined as CONST  

Note that the PI (Procedure Interface) should match the PR (Prototype) exactly in the data type and size of each parameter and the return 
value.  The "names" of the parameters in the Prototype are optional (as illustrated in this example). They could also contain documentary 
words that are different from the parameters in the Procedure Interface.   
The names of the parameters in the PI are required and are the actual names that are used for the input parameter values in the program. 
Note that prior to V7.1, when you have a PI in a member, you must have a PR as well. Also note the comment that the best practice for 
prototypes is to put them into a separate source member and /Copy them into the calling program/procedure. More on this later.  

For fixed form declaration coders: 

 D                 PI              
 D  GrossPay                          8S 2 CONST 
 D  Allowances                        8S 2 CONST 
 D  Tax                               8S 2 

 D TotalPay        S                 10P 2 
  
    TotalPay = GrossPay + Allowances; 

    GrossPay = 0;

Basic Program Call Parameter Options
Optional Parameters 
• OPTIONS(*NOPASS) 

✦ This (and any/all subsequent) parameter is optional 
✦ If any parameter is specified as *NOPASS all subsequent parameters in that prototype must 

also be *NOPASS  
✦ We'll examine this one more closely in a moment 

• OPTIONS(*OMIT)  
✦ The parameter can be omitted 
✦ The special value *OMIT is coded on the call place of the parameter 

Other Parameter Options 
• OPTIONS(*VARSIZE) 

✦ Applies to character fields and arrays passed by reference only 
✦ Allows the parameter to be smaller than specified in the prototype 

• OPTIONS(*RIGHTADJ) - valid only with CONST 
✦ When making a copy of the parameter, right adjusts the content 
✦ i.e. It performs an EVALR when copying the data 

• OPTIONS(*TRIM) 
✦ Similar idea but it trims leading and trailing spaces when copying 
✦ i.e. The same effect as if the parameter were coded as %TRIM(parmname)

The OPTIONS parameter and its associated keywords provide some powerful additional facilities. 
The option *NOPASS means that the parameter does not have to be passed on the call. Any parameters following 
that spec must also have *NOPASS specified. When the parameter is not passed to a program or procedure, the 
called program or procedure will simply function as if the parameter list did not include that parameter. The called 
program can determine the number of parameters it received either by using the *PARMS value in the PSDS or 
more simply by using the %PARMS built-in function. 
*OMIT can be used for parameters which are not mandatory but which occur in the middle of a parameter sequence 
and therefore cannot be designated as *NOPASS.  Some system functions use this option but you will probably not 
use it for your own programs.  When the option is specified the special value *OMIT can be supplied as the 
parameter on the call 
Option *VARSIZE gives the compiler permission to accept a character parameter that is shorter in length than the 
prototype specifies.  This option is often used when the length of the passed field is also passed to the called 
program (for example QCMDEXC does this) 
Option *RIGHTADJ is similar to EVALR.  Use it when the receiving program/procedure expects a right adjusted 
string.  Note that it is only valid when used with CONST.  The original content of the sending field will be right 
adjusted when it is being copied to the temporary parameter. 
Option *TRIM is a newcomer and ensures that leading and trailing spaces are always trimmed from the parameter 
when a copy is being made.

    Dcl-S CusMastOvr Char(40) Inz('OVRDBF FILE(CustMast) .......'); 
    Dcl-S FileName   Char(10); 
    Dcl-S NameToUse  Char(10); 

    Dcl-Pr OvrDBFile ExtPgm('QCMDEXC'); 
      CmdString   Char(3000)   Options(*VarSize) Const; 
      CmdLength   Packed(15:5) Const; 
      CmdOpt      Char(3)      Options(*NoPass) Const; 
    End-Pr;    
                           
    OvrDBFile(CusMastOvr: %Len(CusMastOvr)); 

    OvrDBFile('OVRDBF FILE(' + FileName + ') TOFILE('  
              + NameToUse + ')' : 65); 

Prototype Anything: QCMDEXC
This Prototype allows you to: 
• Use a more meaningful name - OvrDBFile 
• Use a Variable size parameter - Options(*VarSize)  
• Avoid having to move values to 15,5 fields - Const 
• and omit parameters - Options(*NoPass)

This example of a prototype for QCMDEXC demonstrates a number of the additional benefits of prototyping.   
First it allows us to use a name indicative of the function performed - in this case OvrDBFile. 
Second by using Options(*VarSize) we have made it possible to pass the 40 character parameter CusMastOvr when the 
prototype specifies that the length should be 3,000 characters long.  That's 2,940 bytes saved! 
Third we don't have to create a 15,5 work field just so that we can pass an integer length! 
Has that ever struck you as being rather odd?.  If you don't know why QCMDEXC has this rather strange requirement and 
want to know, ask your instructor. 
Lastly we can safely omit the third parameter  
You did know that QCMDEXC has three parameters, didn't you? 
If you didn't, don't feel bad - the third one is only used for DBCS (Double Byte Character Set) support. 

For fixed form declaration coders: 

 D CusMastOvr    S         40   Inz('OVRDBF FILE(CustMast) ........ 

 D OvrDBFile     PR              ExtPgm('QCMDEXC') 
       
 D  CmdString            3000    Options(*VarSize) Const                                                                                 
 D  CmdLength              15P 5 Const                  
 D  CmdOpt                  3    Options(*NoPass) Const  

    OvrDBFile(CusMastOvr: %Len(CusMastOvr)); 

    OvrDBFile('OVRDBF FILE(' + FileName + ') TOFILE('  
              + NameToUse + ')' : 65); 

 
© Copyright Partner400, 2018.

 
Page 11



    Dcl-Pr CalcSalesTax ExtPgm('UTIL0045R'); 
      Amount   Packed(7:2) Const; 
      Tax      Packed(5:2); 
      OptCnty  Char(15)    Const Options(*NoPass); 
      OptState Char(2)     Const Options(*NoPass); 
    End-Pr;               
        
    CalcSalesTax(ItemAmt : TaxAmt : 'Dekalb' : 'GA' ); // Call with 4 parms 
    CalcSalesTax(ItemAmt : TaxAmt : 'Dekalb');       // or with 1st 3 parms 
    CalcSalesTax(ItemAmt : TaxAmt);                  // or with 1st 2 parms

Parameter OPTIONS(*NOPASS)  
Indicates that the parameter is optional 
• ALL subsequent parameters must also be *NOPASS  

Called program (or procedure) checks the number actually passed using %Parms 
• Sample code to follow 

Useful when the called code can provide default values  
• Example: call sales tax routine, optionally including the state & county 

✦ If state and/or county are not supplied, the called routine supplies default values

Note that *NOPASS can only be used when all remaining parameters are also optional and specified as *NOPASS.  
This is similar to the CL command interface where many parameter values are optional. 
Be careful not to confuse *NOPASS with *OMIT - *OMIT does NOT allow you to omit the parameter - it allows you to 
pass *OMIT instead - confusing isn't it! 
*NOPASS allows a parameter to be omitted completely.  The programmer must ensure that the parameter in 
question is not referenced if it has not been passed.  %Parms can be used to establish if the parameter was passed 
or not. 

For fixed form declaration coders: 

 D CalcSalesTax  Pr                    ExtPgm('UTIL0045R') 
 D  Amount                        7P 2 Const   
 D  Tax                           5P 2 
 D  OptCnty                      15A   Const Options(*NoPass)  
 D  OptState                      2A   Const Options(*NoPass) 

        
    CalcSalesTax(ItemAmt : TaxAmt : 'Dekalb' : 'GA' ); // Call with 4 parms 
    CalcSalesTax(ItemAmt : TaxAmt : 'Dekalb');       // or with 1st 3 parms 
    CalcSalesTax(ItemAmt : TaxAmt);                  // or with 1st 2 parms

   Dcl-Pi *N; 
     Amount   Packed(7:2) Const; 
     Tax      Packed(5:2); 
     OptCnty  Char(15)    Const Options(*NoPass); 
     OptState Char(2)     Const Options(*NoPass); 
   End-Pi; 

   Dcl-S Cnty Char(15) INZ('Olmsted'); 
   Dcl-S State Char(2) INZ('MN');    

   If %Parms >= 3; 
      Cnty = OptCnty; 
   EndIf; 
   If %Parms >= 4;  
      State = OptState; 
   EndIf; 

    // Logic must refer only to standalone fields State & Cnty 
    //   Reference to OptCnty or OptState if not passed causes 
    //     run time error ... or much worse ...

Called Routine with *NoPass 
Cardinal rule: Never use a parm that was not passed to you 
• Use %Parms to determine how many you received 

✦ which determines which ones you can use and not use

Called Pgm: 
UTIL0045R

Always, always, always use %Parms to check the number of parms passed if optional parameters are in use. Then never, 
never, never refer to a parameter that was not passed!  
As shown in the example above, If the parameter is not passed, the called routine can handle this by creating a local 
variable which is initialized to the default value.  If the parameter is passed it is moved to this variable which is 
subsequently used in all further processing. 
There are of course other ways of handling this situation  - one of which is simply to condition the logic based on the 
number of parameters received. This would be a valid approach if the parameter was only referenced once in the code. 
Note the big warning at the end of the example - if you refer to or touch a parameter that was not passed to you, you may 
get a pointer exception error - if you're lucky!  If this is a called *PGM, it will be an error.  
But if this were a subprocedure, in many cases there will be no runtime error to halt the routine. The routine will simply use 
the wrong data and potentially update the wrong data, likely corrupting data in another procedure in the job!   
Some of you may have noticed a potential logical flaw in the way this example handles the "default values". Due to limited 
space on the chart, the technique shown here has limitations. See the notes on the next chart for a suggestion of a better 
technique.  
Note: %PARMS relies on the presence of what are known as "minimal descriptors" - RPG IV and CLLE programs always 
pass these.  C and COBOL programs do not do unless told to.  If you are running a mixed language environment and 
intend to use optional parameters, remember to specify descriptors when compiling COBOL and/or C programs. 
For fixed form declaration coders: 

 D CalcSalesTax  PI                      
 D  Amount                        7P 2 Const   
 D  Tax                           5P 2 
 D  OptCnty                      15A   Const Options(*NoPass)  
 D  OptState                      2A   Const Options(*NoPass) 

 D Cnty           S              15A   INZ('Olmsted') 
 D State          S               2A   INZ('MN') 

   Dcl-Pi *N; 
     Amount   Packed(7:2) Const; 
     Tax      Packed(5:2); 
     OptCnty  Char(15)    Const Options(*NoPass); 
     OptState Char(2)     Const Options(*NoPass); 
   End-Pi; 

   Dcl-S Cnty Char(15) INZ('Olmsted'); 
   Dcl-S State Char(2) INZ('MN');    

   If %Parms >= %ParmNum(OptCnty); 
      Cnty = OptCnty; 
   EndIf; 
   If %Parms >= %ParmNum(OptState);   
      State = OptState; 
   EndIf; 

    // Logic must refer only to standalone fields State & Cnty 
    //   Reference to OptCnty or OptState if not passed causes 
    //     run time error ... or much worse ...

Better Version of *NoPass Logic V7 +  
%ParmNum() allows you to soft code parm position 
• %ParmNum(OptCnty) returns 3, %ParmNum(OptState) returns 4 

✦ What if we later added another required parm? 

Called Pgm: 
UTIL0045R

Beginning with V7.1, we have an even better way to compare to %Parms. We can use %ParmNum to compare with the 
%Parms value.  
Suppose we decided at some point in the future to add another required parameter to the parameter list. In that case, we'd need 
to make changes to this logic where it's testing %Parms for a hard-coded position number (e.g., 3 or 4 in our example) because 
required parms must go before the *NoPass ones.  
Using %ParmNum, we can have our logic refer to the position of a specific PI parameter and check that against %Parms to tell 
whether that particular value was passed or not. In that way, the logic continues to work without change, no matter how many 
additional optional parms may be added in the future.  
Of course the example as coded here may have a logical flaw - the "default values" are not constant. Therefore if LR is not set 
on in this program, on a subsequent call, the "defaults" may  now be set to the values used from a previous call. A better 
technique (but too much code for my chart!) might be something like the following logic for each optional parm:  

   If %Parms >= %ParmNum(OptCnty); 
      Cnty = OptCnty; 
   Else 
       Cnty = DftCnty;     // DftCnty is Named Constant with value 'Olmsted' 

For fixed form declaration coders: 

 D CalcSalesTax  PI                      
 D  Amount                        7P 2 Const   
 D  Tax                           5P 2 
 D  OptCnty                      15A   Const Options(*NoPass)  
 D  OptState                      2A   Const Options(*NoPass) 

 D Cnty           S              15A   INZ('Olmsted') 
 D State          S               2A   INZ('MN') 

    Dcl-Pr  CalcSalesTax   ExtPgm('UTIL0045R'); 
      Amount   Packed(7:2) Const; 
      Tax      Packed(5:2); 
      OptCnty  Char(15)    Const Options(*NoPass : *Omit); 
      OptState Char(2)     Const Options(*NoPass); 
    End-Pr;        
        
    // *Omit allows Call with 4 parms, but 3rd is *Omit special value 

    CalcSalesTax(ItemAmt : TaxAmt : *Omit : 'GA' );  

Parameter Options(*Omit)
Allows the special value *OMIT to be used as the parm 
• Called code will check for a null pointer 
• Can be used in conjunction with *NoPass 

✦ Even on the same parameter 

Mostly used when calling certain system APIs 
• But what if it was valid to pass a State but not County?  

The option *OMIT can be used for parameters which are not mandatory but which occur in the middle of a parameter 
sequence and therefore cannot be designated as *NOPASS.   
*OMIT is only allowed for parameters that are passed by reference (not passed by Value), including those with the 
CONST keyword.  More on this in a moment. 
Several system APIs use this option - the parameter is designated as "can be omitted" 
When the option is specified you can either pass an appropriate parameter or the special value *OMIT. 

For fixed form declaration coders: 

 D CalcSalesTax  PR                    ExtPgm('UTIL0045R') 
 D  Amount                        7P 2 Const   
 D  Tax                           5P 2 
 D  OptCnty                      15A   Const Options(*NoPass : *Omit)  
 D  OptState                      2A   Const Options(*NoPass) 
        
    // *Omit allows Call with 4 parms, but 3rd is *Omit 

    CalcSalesTax(ItemAmt : TaxAmt : *Omit : 'GA' );  

   Dcl-Pi *N; 
     Amount   Packed(7:2) Const; 
     Tax      Packed(5:2); 
     OptCnty  Char(15)    Const Options(*NoPass : *Omit); 
     OptState Char(2)     Const Options(*NoPass); 
   End-Pi; 

   Dcl-S Cnty Char(15) INZ('Olmsted'); 
   Dcl-S State Char(2) INZ('MN');    

   If %Parms >= %ParmNum(OptCnty) and %Addr(OptCnty) <> *Null; 
      Cnty = OptCnty; 
   EndIf; 

   If %Parms >= %ParmNum(OptState);  
      State = OptState; 
   EndIf;

Coding for Options(*OMIT) 
Use %Addr(parm_name) to see if it is *Null 
• Sadly we can't check for a value of *Omit 
• Note: Omitted parm is still counted in the number returned by %Parms 

✦ So  %Parms for the following call would return 4 
✦ So must check %Parms (because of *NoPass) and also check for *Null (because of *Omit) 

• So the caller may code this call if State is required, but Cnty not required 
          CalcSalesTax( ItemAmt : TaxAmt : *Omit : 'GA' );  

Called Pgm: 
UTIL0045R

Note that the parameter will still count in the number of parameters passed and the called program will need to test 
to see if the parameter was actually passed.  Any attempt to reference the parameter when *OMIT was passed will 
result in an error. There are two ways to do this: 
The primary/easiest way is to compare the %Addr of the parameter to *Null special value. 
You can also use the API CEETSTA  (The RPG Programmers Guide provides a brief example.) to determine if a 
parameter was actually passed. 

For fixed form declaration coders: 

 D CalcSalesTax  PI                     
 D  Amount                        7P 2 Const   
 D  Tax                           5P 2 
 D  OptCnty                      15A   Const Options(*NoPass : *Omit)  
 D  OptState                      2A   Const Options(*NoPass) 

 D Cnty           S              10A   INZ('Olmsted') 
 D State          S               2A   INZ('MN') 

   If %Parms >= %ParmNum(OptCnty) and %Addr(OptCnty) <> *Null; 
      Cnty = OptCnty; 
   EndIf; 
   If %Parms >= %ParmNum(OptState);  
      State = OptState; 
   EndIf;

       Dcl-Pr DayOfWeek   Zoned(1); 
         ADate  Date(*USA)  Value; 
       End-Pr;      
        
       DayNumber = DayofWeek(InputDate);

Prototypes for Bound Procedures 
Simply omit the EXTPGM parameter 
• Default is EXTPROC - not required unless actual procedure name is 

different from Prototype name (translated to upper case) 
✦ Remember - Program and Procedure names are case sensitive 

Return value may be specified 
• In example below, a 1 digit number is returned 
• Note the way the procedure is called 

Parameters may be passed by VALUE 
• A copy of the actual data is passed to the called procedure 

✦ Not a pointer to the data as when passing by reference 
✦ Used most often in the C language 

- More on calling C functions shortly 

We've concentrated here on prototypes for program calls more than for bound procedures, be they RPG 
subprocedures or bindable system APIs or other procedure calls.  

Note that that procedure name does not have to be specified unless it's name is something other than an all upper-
case version of the prototype name. That's what the default will be.  

In addition to the ability to pass parameters by VALUE as shown on this chart, one other option is also only available 
to bound procedure prototypes - the ability to use OPTIONS(*String). This option is especially useful for passing to C 
functions and is most often used with *TRIM. It automatically places null terminator at end of the value passed so the 
programmer doesn't need to do this work manually.  

For fixed form declaration coders: 

     D DayOfWeek       PR             1S 0                                                                      
     D  ADate                          D    Value DatFmt(*USA)    
        
       DayNumber = DayofWeek(InputDate); 

Who Should Write the Prototype?
If *PGM A calls *PGM B, who should write the PR? 
• i.e., the programmer responsible for *PGM A or the one for *PGM B? 
• Ideally, the programmer responsible for *PGM B  

✦ Because that person best knows how the parameters are used 
✦ e.g., whether or not CONST or *NOPASS are viable options 

Therefore, the *PGM B programmer should write the PR 
• In a separate member that is /Copy'd the PR into *PGM B 

✦ The compiler will check it against the PI in *PGM B  
- Which should, of course, also be present! 

• That way the same "verified" PR can safely /Copy'd into all callers 
• That's why PRs are typically written in separate source members 

✦ Sometimes multiple PRs are combined into a single /Copy member 
Note: Prototypes do not occupy space in a compiled program 
• So it does not affect the size or performance of a *PGM if a member 

with many PRs not used in this program is copied in 
✦ Tho it may cause other issues, such as "clutter" in your Outline View in RDi 

Most shops who use prototypes make extensive use of /Copy to bring those prototypes into the calling programs 
(and the called program as well).  This chart explains the reasoning behind this common (and recommended) 
practice.

New Prototype Rules for 7.1+
Prior to 7.1, the compiler required:  
• A matching prototype must be present in any member when a PI exists 
• V7.1 it is technically no longer necessary in most cases, however it is still 

a best practice for external procedures and called programs 
Examples of when a prototype isn't necessary as of 7.1  

✦ An internal procedure (not callable externally)  
✦ A program or procedure will never be called by another RPG pgm/module 

- e.g. If only called from CL, then the prototype will never be used 
✦ One used only as an exit program or command-processing program for command  

Otherwise, even if the compiler doesn't require it, code a prototype 
✦ Put it into a separate source member 
✦ /Copy it into the member where PI is - i.e., where the program or procedure code is 

- Because if you put the PR & the PI in the same source member, the compiler will validate 
the PR to ensure it is correct 

✦ Then /Copy that PR into all *PGMs or procs that call this one to ensure correct calls 

This is the best practice 
✦ When in doubt of the future of the code, code a PR anyway

I intend to still put my prototypes in the source members where the program or subprocedures are coded unless I'm 
absolutely confident it will never be called from another RPG Program. That way the compiler can check them for 
me. And I'll still put them in /Copy members so that I'm sure the version that has been checked is used by all the 
callers of that routine. 

There are 2 primary exceptions: 
1) If I have a program (or less likely, a procedure) that is never likely to be called from another RPG program - e.g., if 
it is only called from a CL  
2) Internal procedures  - aka procedures that are subroutine alternatives. If the procedures are not exported, they 
cannot be called from outside the module where they are coded. So there is no need for a prototype in this case 
after V7.1. Prior to V7.1, the compiler requires it. 

Before moving on to Prototyping C 
functions and C-Style APIs . . . 

Any questions about prototyping 
your RPG code? 

   Dcl-Pr  Sine Float(8) ExtProc('sin'); 
     Double  Float(8)  Value; 
   End-Pr; 

   Dcl-Pr Cosine Float(8) ExtProc('cos'); 
     Double  Float(8)  Value; 
   End-Pr;    
        

   // Once defined these functions can be used as if they were built into RPG IV. 
        

   SineX   = Sine(X) 

   CosineY = Cosine(Y); 

Using Simple C Math Functions
Like most C functions the parameter must be passed by value 
• So the actual content (i.e. the value) of the parameter field is passed 

Here are the prototypes for the Sine and Cosine functions 
• Both return a double length floating point value (8F) 

C defines the prototype as:   
• double sin(double x)  and  double cos(double x) 

Question: 
• What data type should the variables X and SineX be?

Notes
The C math functions include acos, asin, atan, cos, cosh, sin, sinh, tan, tanh.  All accept double floating point 
parameters (8F in RPG IV) passed by value, and all return a double floating point value. 

It may seem to you that these kinds of math functions are rarely used in RPG programs, and you are right.  However, 
on those occasions when they are needed then you will really appreciate the fact that you don't have to try and 
"fake" the function out in RPG. 

The RPG Redbook (referenced later in this handout) includes an example of these C functions in action.  

     D Sine            PR             8F   ExtProc('sin') 
     D  Double                        8F   Value 

     D Cosine          PR             8F   ExtProc('cos') 
     D  Double                        8F   Value 

H  DftActGrp(*No) 

  // or  

  Ctl-Opt  DftActGrp(*No); 

Is that all there is to it ?
Not quite - there is one more thing we need to do 
• We must specify that the program is a "real" ILE one 
• We do this by specifying DFTACTGRP(*NO) 

✦ This is assuming we're compiling using CRTBNDRPG command/option 
There is no need to worry about which Service Programs to use 
• The system binding directory takes care of that 

✦ Unless you are on a really old system (i.e. pre V6) 
- If so, use binding directory QC2LE  

Embedding the parameters on the control/H-spec works well 
• You won't forget them that way !

Notes
C function calls are bound calls and can only be used from "real" ILE programs, as opposed to the OPM compatible 
programs produced by DFTACTGRP(*YES).  So we have to specify DFTACTGRP(*NO) 
Note that if you are compiling the program with the CRTBNDRPG command, when you first prompt the command 
you will not see the Binding Directory parameter.  This will not appear until you have entered *NO for the 
DFTACTGRP parameter - not even if you press F10. 
IBM supplies a System Binding Directory and one for each compiler.  These are used to allow the Binder to locate 
the required run-time routines.  The compiler automatically passes the name of its own Binding Directory to the 
Binder, there is no need for the programmer to specify it.   
The C compiler's Binding Directory is (you guessed it!) QC2LE.  There is no need to specify unless you are on a 
release prior to V6. 
As usual, if we want to run the program in an Activation group other than QILE then the ACTGRP keyword must also 
be specified. 

    Dcl-Pr sleep Int(10) extproc('sleep'); 
      seconds Uns(10) value; 
    End-Pr; 

    Dcl-Pr usleep Int(10) extproc('usleep'); 
      microsecs Uns(10) value; 
    End-Pr;     
     
    // Sleep for ten seconds 
    sleep(10); 

    // Sleep for one tenth of a second 
    usleep(100000)

Two Useful C Library Functions
sleep and usleep 
• sleep puts the program to sleep for a specified number of seconds 
• usleep puts it to sleep for a number of microseconds 

Note that in the examples the return value is ignored 
• i.e. We are effectively using CALLP not EVAL 

Return value would be -1 if the request failed and 0 if it worked 
• Not terribly useful information - so it is often ignored

Notes
Want your RPG program to take a little nap while it waits to see if a Element lock is released?  Or perhaps to snooze 
while waiting for a file in the IFS to be available for processing? 

Meet sleep and usleep - these are among the C functions classified as part of the UNIX type APIs.  Also included in 
this set are the APIs that we use to process IFS files, among other things.

  D sleep           PR            10I 0 extproc('sleep') 
  D   seconds                     10U 0 value 

  D usleep          PR            10I 0 extproc('usleep') 
  D   microsecs                   10U 0 value 

   /Free 
    // Sleep for ten seconds 
    sleep(10); 

    // Sleep for one tenth of a second 
    usleep(100000) 

       // Proto for CEERan0 
       //   Seed range is 0 to 2,147,483,646. Set from GMT if zero 
       //    - If returned RandomNumber is -1 then seed was invalid 
       //   API sets seed to new value ready for the next number  

       Dcl-Pr GetRandomNum ExtProc('CEERAN0'); 
         Seed         Int(10); 
         RandomNumber Float(8); 
         Feedback     Char(12)  Options(*Omit); 
       End-Pr; 

       Dcl-S Seed      Int(10) Inz; 
       Dcl-S FloatRand Float(8) Inz; 
       Dcl-S Feedback  Char(12);  

       // Call GetRandomNum to obtain the random number (a float in range 0 - 1) 

       GetRandomNum( Seed : FloatRand : *Omit );

Random Number Generation
The basic C function for random number generation is rand() 
• But it has limited range and there are better options 

CEERAN0 works well 
• It returns a floating point value in the range 0 to 1 
• Then simply multiply this by the size of the range of numbers you want

Notes
While it may seem odd that the number returned only ranges from 0 to 1, it makes it very easy to generate numbers 
in a required range. 
Assume you want numbers in the range 50 to 199 then you need one of 150 possible numbers. If we place 50 in the 
variable LowVal and 199 in the variable HighVal then the following calculation will give us a random number within 
that range. 
 ( RandomNumber * ( HighVal - LowVal ) ) + LowVal;  
e.g. If RandomNumber = 0 (the bottom of the possible range) the sum is: 
( 0 * (199 - 50) ) + 50 - which results in a value of 50. 
If RandomNumber = 1 (the top of the range) then the sum is: 
( 1 * (199 - 50) ) + 50 - which results in 199. Voila!

       // Proto for CEERan0 
       //   Seed range is 0 to 2,147,483,646. Set from GMT if zero 
       //    - If returned RandomNumber is -1 then seed was invalid 
       //   API sets seed to new value ready for the next number  

     D RandomNum       Pr                  ExtProc('CEERAN0') 
     D  Seed                         10I 0 
     D  RandomNumber                  8F 
     D  Feedback                     12A   Options(*Omit) 

     D Seed            S             10I 0 Inz 
     D FloatRand       S              8F   Inz 
     D Feedback        S             12A 

   Dcl-S command       Char(40) Inz('CL command goes here'); 

   Dcl-Pr CallSystem   Int(10)  ExtProc('system'); 
     CmdString         Pointer  Value  Options(*String: *Trim); 
   End-Pr;                 

   Reply =  CallSystem(command);       

How to deal with C strings
Many C-style function require the use of string parameters 
• But C strings are variable in length and null terminated 

✦ i.e. The last valid character is followed by hex zeros (X'00') 
• They are also passed as pointers 

Options(*String) allows you to pass a regular RPG fixed length field 
• Compiler takes care of the null termination and creating the pointer etc. 

Adding the *TRIM option makes it easier to use fixed length fields 
• Without it the safest way is to pass a varying length field  

Example: the C "system" API - an alternative to QCMDEXC 
• It gives you a simple way to test if the operation was successful 

Notes
Many published examples of using C type functions fail to use prototyping, or fail to use it correctly and you will often see 
code like the following: 

     D Command         S             40a   Inz('Command string goes here') 

     D TempCommand     S             41a 
      *....................................      
     D CallSystem      Pr            10i 0 ExtProc('system') 
     D   CmdString                     *   Value 

       TempCommand = %Trim(Command) + X'00'; 
       Reply = CallSystem(%Addr(TempCommand)) 

As you can see this is similar to our example above - but requires more coding and the use of a temporary variable.  Since 
Options(*String) was introduced in V3R7 - only one release after it became possible to call C functions at all it is surprising 
that some many "wrong" examples are out there. Our take is that if the compiler writers give us a better solution - we use it!

 D command         S             40a   Inz('CL command goes here') 

           
 D CallSystem      Pr            10i 0 ExtProc('system') 
 D   CmdString                     *   Value Options(*String: *Trim) 
  

   Reply = CallSystem(command); 

   Dcl-Pr CallSystem Int(10)    ExtProc('system'); 
     CmdString  Pointer  Value  Options(*String); 
   End-Pr; 
                                                       
   Reply = CallSystem(Command); 
           
    If Reply <> 0; 
      Dsply ('Reply was ' + %Char(Reply)); 
      Dsply ('_EXCP_MSGID = ' + ErrorCode);           
    EndIf;    

More on using the C system function
system returns a non-zero number in the event of failure 
• You can also determine the message was that issued 
• It is contained in the IMPORTed field _EXCP_MSGID 

✦ But this is the only information available  

Notes
Note here the use of the Options(*String) on the prototype.  This allows us to use a conventional RPG string as the 
parameter to the function and compiler generated code will take care of copying it to a null-terminated string and 
passing a pointer to that string to the C function. 
The returned value is a four byte integer (10 I) which will be set negative (-1) in the event of an error.   
If there is an error, the field ErrorCode will contain the conventional 7 character IBM i message code.  This is achieved 
by mapping the field through the keyword Import and specifying the C name for the field.  i.e. Import('_EXCP_MSGID')  
C programmers often use and test the results of a function like this in one step.  In our example, we are reporting the 
actual value of the reply code and therefore need to invoke the function and test the result separately.  If this was not a 
requirement (and it normally isn't) the two lines: 
   Reply = CallSystem(Command); 
   If Reply <> 0; 
Could have been replaced by the simple: 
   If CallSystem(Command) <> 0; 
  

 D Reply           S             10I 0 
 D Command         S             40A   Inz('Command goes here') 
           
 D CallSystem      Pr            10I 0 ExtProc('system') 
 D   CmdString                     *   Value Options(*String) 

 D ErrorCode       S              7    Import('_EXCP_MSGID') 

   // Prototype for __errno() - Retrieve pointer to C errno 
 Dcl-Pr GetErrno Pointer Extproc('__errno'); 
 End-Pr;      
                                                         
  // ErrorNo is BASED on pointer returned by __errno() 
 Dcl-S ErrorNo Int(10) Based(pErrorNo);  

                                                      
   pErrorNo  = GetErrno(); 
     // The actual error number is now available in the ErrorNo variable  

Handling C-style function errors
If the function signals an error ( -1 ) you need to check errno 
• But you cannot access this variable directly  

C-style functions set it to identify the specific error 
• A list of values used by a specific API are described in its docs  
• The list of all the possible errno values can be found in the manuals for 

"UNIX-Type APIs - Integrated File System (IFS) APIs" 
To access the value use the function __errno()  
• We will look at its companion function strerror() in a moment 
• Note that there are two underscores in __errno

    // Prototype for strerror() - Retrieve error text for given errno 

   Dcl-Pr StrError Pointer ExtProc('strerror'); 
     errno Int(10) Value; 
   End-Pr;          
                                                          
    // Variables for strerror() function 
   Dcl-S pErrorTxt Pointer; 
   Dcl-S message Char(52);                                                                 
   
   pErrorTxt = StrError(errno); 
   message = 'Error code: ' + %char(ErrorNo) + ': ' 
                + %Str(pErrorTxt); 
   dsply message;

Handling C-style function errors - contd.
There are text descriptions associated with each errno value 
The function strerror() can be used to retrieve the error text 
• Pass it the errno value to retrieve the actual error message text 

✦ In this example we will be truncating anything beyond the first 30 or so characters 
• Note that strerror() returns a pointer to a C-style null terminated string 

✦ So we need to use %STR to handle it as a conventional RPG variable  

